Math 173
Calculus I

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 5
The Definition of the Derivative
The diff and D commands

Initializations

```
> restart;
```

5.1 The definition of the derivative

As explaned in class, the derivative of the function $f(x)$ is defined as: $\lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$.

Examples

Example 5.1.1

Use the definition of the derivative to compute the derivative of $f(x) = \frac{5}{\sqrt{3-2x}}$.

Solution
First define the function, then execute the limit.

```
> f:=x->5/sqrt(3-2*x);
```

$$f := x \to \frac{5}{\sqrt{3-2x}} \quad \textbf{(2.1.1.1)}$$

```
> e1:=Limit((f(x+h)-f(x))/h, h=0);
```

$$e1 := \lim_{h \to 0} \left(\frac{\frac{5}{\sqrt{3-2x-2h}} - \frac{5}{\sqrt{3-2x}}}{h} \right) \quad \textbf{(2.1.1.2)}$$

```
> der1:=value(e1);
```

$$der1 := \frac{5}{(3-2x)^{3/2}} \quad \textbf{(2.1.1.3)}$$

```
>
```

5.2 The diff and D routines

Maple's **diff** command allows the user to seemlessly compute the derivative of given expression.

Examples

Example 5.2.1

Compute the derivative of the function $f(x) = x \sin x$.

Solution

```
> f:=x->x*sin(x);
```

$$f := x \to x \sin(x) \tag{3.1.1.1}$$

First we apply the **diff** command to the expression $f(x)$.

```
> der1:=diff(f(x), x);
```

$$der1 := \sin(x) + x\cos(x) \tag{3.1.1.2}$$

Alternatively, the **D** routine may be used. It applies to the Maple function *f*.

```
> der1:=D(f)(x);
```

$$der1 := \sin(x) + x\cos(x) \tag{3.1.1.3}$$

Use of the **D** routine is preferred when the derivative needs to be evaluated for a specific value of the argument. For instance, $f'(\pi)$ is found as follows.

```
> D(f)(Pi);
```

$$-\pi \tag{3.1.1.4}$$

Example 5.2.2

Compute the equation of the tangent line at the point with x-coordinate 3 to the curve $f(x) = \dfrac{1}{\sqrt{2x+3}}$ **. Show the curve and the tangent line in one picture.**

Solution

Code the function.

```
> f:=x->1/sqrt(2*x+3);
```

$$f := x \to \frac{1}{\sqrt{2x+3}} \tag{3.1.2.1}$$

```
> tangent_line:=y-f(3)=D(f)(3)*(x-3);
```

$$tangent\_line := y - \frac{1}{3} = -\frac{1}{27}x + \frac{1}{9} \tag{3.1.2.2}$$

To obtain the tangent line as an expression in *x* we solve the equation above for *y*.

```
> TL_x:=solve(tangent_line, y);
```

$$TL\_x := \frac{4}{9} - \frac{1}{27}x \tag{3.1.2.3}$$

```
> plot({f(x), TL_x}, x=0..6);
```


0.55
0.50
0.45
0.40
0.35
0.30
0.25
0
1
2
3
4
5
6
x

>